Complete Non-Orders and Fixed Points

Akihisa Yamada
National Institute of Informatics, Tokyo, Japan
akihisayamada@nii.ac.jp

Jérémy Dubut
National Institute of Informatics, Tokyo, Japan
Japanese-French Laboratory for Informatics, Tokyo, Japan
dubut@nii.ac.jp

Abstract

In this paper, we develop an Isabelle/HOL library of order-theoretic concepts, such as various completeness conditions and fixed-point theorems. We keep our formalization as general as possible: we reprove several well-known results about complete orders, often without any property of ordering, thus complete non-orders. In particular, we generalize the Knaster–Tarski theorem so that we ensure the existence of a quasi-fixed point of monotone maps over complete non-orders, and show that the set of quasi-fixed points is complete under a mild condition—attractivity—which is implied by either antisymmetry or transitivity. This result generalizes and strengthens a result by Stauti and Maaden. Finally, we recover Kleene's fixed-point theorem for omega-complete non-orders, again using attractivity to prove that Kleene's fixed points are least quasi-fixed points.

2012 ACM Subject Classification Theory of computation → Interactive proof systems

Keywords and phrases Order Theory, Lattice Theory, Fixed-Points, Isabelle/HOL

Digital Object Identifier 10.4230/LIPIcs.ITP.2019.13

Related Version Akihisa Yamada and Jérémy Dubut. Complete Non-Orders and Fixed Points. *Archive of Formal Proofs*, 2019. (`http://isa-afp.org/entries/Complete_Non_Orders.html`, Formal proof development)

Funding The authors are supported by ERATO HASUO Metamathematics for Systems Design Project (No. JPMJER1603), JST; J. Dubut is also supported by Grant-in-aid No. 19K20215, JSPS.

1 Introduction

The main driving force towards mechanizing mathematics using proof assistants has been the reliability they offer, exemplified prominently by [10], [12], [14], etc. In this work, we utilize another aspect of proof assistants: they are also engineering tools for developing mathematical theories. In particular, we choose Isabelle/JEdit [22], a *very* smart environment for developing theories in Isabelle/HOL [17]. There, the proofs we write are checked "as you type", so that one can easily refine proofs or even theorem statements by just changing a part of it and see if Isabelle complains or not. Sledgehammer [7] can often automatically fill relatively small gaps in proofs so that we can concentrate on more important aspects. Isabelle's counterexample finders [3, 6] should also be highly appreciated, considering the amount of time one would spend trying in vain to prove a false claim.

In this paper, we formalize order-theoretic concepts and results in Isabelle/HOL. Here we adopt an *as-general-as-possible* approach: most results concerning order-theoretic completeness and fixed-point theorems are proved without assuming the underlying relations to be orders (non-orders). In particular, we provide the following:

- Various completeness results that generalize known theorems in order theory: Actually most relationships and duality of completeness conditions are proved without *any* properties of the underlying relations.

10th International Conference on Interactive Theorem Proving (ITP 2019).
Editors: John Harrison, John O'Leary, and Andrew Tolmach; Article No. 13; pp. 13:1–13:16
Leibniz International Proceedings in Informatics
LIPICS Schloss Dagstuhl – Leibniz-Zentrum für Informatik, Dagstuhl Publishing, Germany

- Existence of fixed points: We show that a relation-preserving mapping $f : A \to A$ over a complete non-order $\langle A, \sqsubseteq \rangle$ admits a *quasi-fixed point* $f(x) \sim x$, meaning $x \sqsubseteq f(x) \wedge f(x) \sqsubseteq x$. Clearly if $\sqsubseteq$ is antisymmetric then this implies the existence of fixed points $f(x) = x$.
- Completeness of the set of fixed points: We further show that if $\sqsubseteq$ satisfies a mild condition, which we call *attractivity* and which is implied by either transitivity or antisymmetry, then the set of quasi-fixed points is complete. Furthermore, we also show that if $\sqsubseteq$ is antisymmetric, then the set of *strict* fixed points $f(x) = x$ is complete.
- Kleene-style fixed-point theorems: For an ω-complete non-order $\langle A, \sqsubseteq \rangle$ with a bottom element $\bot \in A$ (not necessarily unique) and for every ω-continuous map $f : A \to A$, a supremum exists for the set $\{f^n(\bot) \mid n \in \mathbb{N}\}$, and it is a quasi-fixed point. If $\sqsubseteq$ is attractive, then the quasi-fixed points obtained this way are precisely the least quasi-fixed points.

We remark that all these results would have required much more effort than we spent (if possible at all), if we were not with the aforementioned smart assistance by Isabelle. Our workflow was often the following: first we formalize existing proofs, try relaxing assumptions, see where proof breaks, and at some point ask for a counterexample.

The formalization is available in the Archive of Formal Proofs.

Related Work

Many attempts have been made to generalize the notion of completeness for lattices, conducted in different directions: by relaxing the notion of order itself, removing transitivity (pseudo-orders [19]); by relaxing the notion of lattice, considering minimal upper bounds instead of least upper bounds (χ-posets [15]); by relaxing the notion of completeness, requiring the existence of least upper bounds for restricted classes of subsets (e.g., directed complete and ω-complete, see [8] for a textbook). Considering those generalizations, it was natural to prove new versions of classical fixed-point theorems for maps preserving those structures, e.g., existence of least fixed points for monotone maps on (weak chain) complete pseudo-orders [5, 20], construction of least fixed points for ω-continuous functions for ω-complete lattices [16], (weak chain) completeness of the set of fixed points for monotone functions on (weak chain) complete pseudo-orders [18].

Concerning Isabelle formalization, one can easily find several formalizations of complete partial orders or lattices in Isabelle's standard library. They are, however, defined on partial orders, either in form of classes or locales, and thus not directly reusable for non-orders. Nevertheless we tried to make our formalization compatible with the existing ones, and various correspondences are ensured in the Isabelle source.

2 Preliminaries

This work is based on Isabelle 2019. In Isabelle/HOL, R :: 'a ⇒ 'a ⇒ bool means a binary predicate R, by which we represent a binary relation $R \subseteq A \times A$. Here A is the universe of the type variable 'a, in Isabelle's syntax, UNIV :: 'a set. Type annotations ":: _" are omitted unless they are necessary. We call the pair $\langle A, \sqsubseteq \rangle$ of a set A and a binary relation $(\sqsubseteq)$ over A a *related set*. One could also call it a *graph* or an *abstract reduction system*, but then some terminology like "complete" become incompatible.

To make our library *as general as possible*, we avoid using the order symbol $\leq$, which is fixed by the class mechanism of Isabelle/HOL. Instead we make the relation of concern explicit as an argument, sometimes called the *dictionary-passing* style [11]. On one hand this design choice adds a notational burden, but on the other hand it allows instantiating obtained results to arbitrary relations over a type, for which the class mechanism fixes one ordering. In the formalization we also import our results into the class hierarchy.

A map $f : I \to A$ over related sets from $\langle I, \preceq \rangle$ to $\langle A, \sqsubseteq \rangle$ is *relation preserving*, or *monotone*, if $i \preceq j$ implies $f(i) \sqsubseteq f(j)$. For this property there already exists a definition in the standard Isabelle library:

```
monotone (≼) (⊑) f ⟷ (∀ i j. i ≼ j ⟶ f i ⊑ f j)
```

Hereafter, in our Isabelle code, we use symbols (⊑) denoting a variable of type 'a ⇒ 'a ⇒ bool, and (≼) denoting a variable of type 'i ⇒ 'i ⇒ bool. More precisely, statements and definitions using these symbols are made in a context such as

```
context fixes less_eq :: "'a ⇒ 'a ⇒ bool" (infix "⊑" 50)
```

For clarity, we present definitions, e.g., of predicates for being upper/lower bounds and greatest/least elements, as

```
definition "bound (⊑) X b ≡ ∀x ∈ X. x ⊑ b"
definition "extreme (⊑) X e ≡ e ∈ X ∧ (∀x ∈ X. x ⊑ e)"
```

making the relation (⊑) of concern as an explicit parameter. Note that we chose such constant names that do not suggest which side is greater or lower. The least upper bounds (suprema) and greatest lower bounds (infima) are thus uniformly defined as follows.

```
abbreviation "extreme_bound (⊑) X ≡ extreme (⊒) {b. bound (⊑) X b}"
```

Hereafter, we write (⊒) for (⊑)⁻, which is also an abbreviation:

```
abbreviation "(⊑)⁻ x y ≡ y ⊑ x"
```

We can already prove some useful lemmas. For instance, if $f : I \to A$ is relation preserving and $C \subseteq I$ has a greatest element $e \in C$, then $f(e)$ is a supremum of the image $f(C)$. Note here that no assumption is imposed on the relations $\preceq$ and $\sqsubseteq$.

```
lemma monotone_extreme_imp_extreme_bound:
  assumes "monotone (≼) (⊑) f" and "extreme (≼) C e"
  shows "extreme_bound (⊑) (f ` C) (f e)"
```

2.1 Locale Hierarchy of Relations

We now define basic properties of binary relations, in form of *locales* [13, 2]. Isabelle's locale mechanism allows us to conveniently manage notations, assumptions and facts. For instance, we introduce the following locale to fix a relation parameter and use infix notation.

```
locale less_eq_syntax = fixes less_eq :: "'a ⇒ 'a ⇒ bool" (infix "⊑" 50)
```

The most important feature of locales is that we can give assumptions on parameters. For instance, we define a locale for reflexive relations as follows.

```
locale reflexive = less_eq_syntax + assumes refl[iff]: "x ⊑ x"
```

This declaration defines a new predicate "reflexive", with the following defining equation:

```
theorem reflexive_def: "reflexive (⊑) ≡ ∀x. x ⊑ x"
```

One may doubt that such a simple assumption deserves a locale not just the definition. Nevertheless, we have some useful lemmas already, for instance:

```
lemma (in reflexive) extreme_singleton[simp]: "extreme (⊑) {a} b ⟷ a = b"
lemma (in reflexive) extreme_bound_singleton[iff]: "extreme_bound (⊑) {a} a"
```

Similarly we define transitivity and antisymmetry:

```
locale transitive = less_eq_syntax + assumes trans[trans]: "x ⊑ y ⟹ y ⊑ z ⟹ x ⊑ z"
locale antisymmetric = less_eq_syntax +
  assumes antisym[dest]: "a ⊑ b ⟹ b ⊑ a ⟹ a = b"
```

It is straightforward to have locales that combine the above assumptions. Some famous combinations are *quasi-orders* for reflexive and transitive relations and *partial orders* for antisymmetric quasi-order.

```
locale quasi_order = reflexive + transitive
locale partial_order = quasi_order + antisymmetric
```

Less known, but still a convenient assumption is being a *pseudo-order*, coined by Skala [19] for reflexive and antisymmetric relations. There, the supremum of a singleton set $\{x\}$ uniquely exists—x itself.

```
locale pseudo_order = reflexive + antisymmetric
```

```
lemma (in pseudo_order) extreme_bound_singleton_eq[simp]:
  "extreme_bound (⊑) {x} y ⟷ x = y" by auto
```

It is clear that a partial order is also a pseudo-order, which is stated by the following *sublocale* declaration. Afterwards facts proved in pseudo_order will be automatically available in partial_order.

```
sublocale partial_order ⊆ pseudo_order..
```

Although these combinations are sufficient for the rest of this paper, we also present all locales combining these basic properties and their relationships in Fig. 1.

3 Completeness of Non-Orders

Here we formalize various order-theoretic completeness conditions in Isabelle. Order-theoretic completeness demands certain subsets of elements to admit suprema or infima. The strongest completeness requires that any subset of elements has suprema and infima.

```
locale complete = less_eq_syntax + assumes "Ex (extreme_bound (⊑) X)"
```

The above assumption only requires suprema (if the right-hand side of ⊑ is seen greater) but not infima, in Isabelle, "Ex (extreme_bound (⊒) X)". This is a well-known consequence in complete lattices, and luckily the proof does not rely on any property of orders. Hence we can declare the following sublocale:

```
sublocale complete ⊆ dual: complete "(⊒)"
proof
  fix X :: "'a set"
  obtain s where "extreme_bound (⊑) {b. bound (⊒) X b} s" using complete by auto
  then show "Ex (extreme_bound (⊒) X)" by (intro exI[of _s] extreme_boundI, auto)
qed
```



Figure 1 Combinations of basic properties. The black dot around the center represents arbitrary binary relations, and the five outgoing arrows indicate atomic assumptions. We do not present the combination of reflexive and irreflexive, which is empty, and one of symmetric and antisymmetric, which is a subset of equality. Node "¬tolerance" indicates the negated relation is tolerance, and "∅" is the empty relation.

Afterwards, a theorem named xxx proved in locale complete will be available in its dual form as dual.xxx.

Let us mention another strong completeness condition: every nonempty subset of elements has a supremum. This condition is called *semicompleteness*, cf. [4, Chapter 6].

```
locale semicomplete = less_eq_syntax +
  assumes "X ≠ {} ⟹ Ex (extreme_bound (⊑) X)"
```

However, semicompleteness fails to be self-dual. Instead, duality holds for a slightly weaker, but highly important completeness condition, *conditional completeness* or *Dedekind completeness*, asserting that any nonempty bounded set has a supremum.

```
locale conditionally_complete = less_eq_syntax +
  assumes "Ex (bound (⊑) X) ⟹ X ≠ {} ⟹ Ex (extreme_bound (⊑) X)"
sublocale conditionally_complete ⊆ dual: conditionally_complete "(⊒)"
```

Let us also mention a very weak form of completeness. A related set $\langle A, \sqsubseteq\rangle$ is called *bounded* if there is a "top" element $\top \in A$, a greatest element in A. Note that there might be multiple tops if $(\sqsubseteq)$ is not antisymmetric.

```
locale bounded = less_eq_syntax + assumes "∃t. ∀x. x ⊑ t"
```

This notion can be also seen as a completeness condition, since it is equivalent to saying that the universe has a supremum.

```
lemma bounded_iff_UNIV_complete: "bounded (⊑) ⟷ Ex (extreme_bound (⊑) UNIV)"
```

Since a top element is a bound of any subset of elements, a conditionally complete relation is semicomplete if (and only if) it is bounded.

```
proposition semicomplete_iff_conditionally_complete_bounded:
  shows "semicomplete (⊑) ⟷ conditionally_complete (⊑) ∧ bounded (⊑)"
```

The dual notion of bounded is called *pointed*. There, a least element is called a "bottom" element, and serves as a supremum of the emptyset. The dual form of the above proposition, together with the duality of conditional completeness means that, (⊒) is semicomplete if and only if (⊑) is pointed conditionally complete. The latter means that every bounded set, including the empty set, has a supremum—the notion known as "bounded complete".

```
proposition bounded_complete_iff_dual_semicomplete:
  "bounded_complete (⊑) ⟷ semicomplete (⊒)"
```

3.1 Lattice-Like Completeness

One of the most well-studied notion of completeness would be the semilattice condition: every pair of elements x and y has a supremum $x \sqcup y$ (not necessarily unique if the underlying relation is not antisymmetric).

```
locale pair_complete = less_eq_syntax + assumes "Ex (extreme_bound (⊑) {x,y})"
```

It is well known that in a semilattice, i.e., a pair-complete partial order, every finite nonempty subset of elements has a supremum. We prove the result assuming transitivity, but only that.

```
locale finite_complete = less_eq_syntax +
  assumes "finite X ⟹ X ≠ {} ⟹ Ex (extreme_bound (⊑) X)"

locale trans_semilattice = transitive + pair_complete

sublocale trans_semilattice ⊆ finite_complete
```

Proof. The proof is an easy induction on the finite set X. Only a care is taken for the case where X is singleton $\{x\}$; then x may fail to be a supremum of itself, as we do not have reflexivity. Instead we find a supremum via that of the pair of x and x. ◀

3.2 Directed Completeness

Directed completeness is an important notion in domain theory [1], asserting that every nonempty directed set has a supremum. Here, a set X is *directed* if any pair of two elements in X has a bound in X.

```
definition "directed (⊑) X ≡ ∀x ∈ X. ∀y ∈ X. ∃z ∈ X. x ⊑ z ∧ y ⊑ z"
```

```
locale directed_complete = less_eq_syntax +
  assumes "directed (⊑) X ⟹ X ≠ {} ⟹ Ex (extreme_bound (⊑) X)"
```

The image of a relation-preserving map preserves directed sets.

```
lemma monotone_directed_image:
  assumes "monotone (⪯) (⊑) f" and "directed (⪯) D" shows "directed (⊑) (f ‘ D)"
```

Gierz et al. [9] showed that a directed complete partial order is semicomplete if and only if it is also a semilattice. We generalize the claim so that the underlying relation is only transitive.

```
proposition (in transitive) semicomplete_iff_directed_complete_pair_complete:
  shows "semicomplete (⊑) ⟷ directed_complete (⊑) ∧ pair_complete (⊑)"
```

Proof. The $\longrightarrow$ direction is trivial. For the other direction, consider a nonempty set X. We collect all suprema of every nonempty finite subset Y of X into a set S:

$$S = \{x.\ \exists Y \subseteq X.\ \mathsf{finite}\ Y \wedge Y \neq \{\} \wedge \mathsf{extreme\_bound}\ (\sqsubseteq)\ Y\ x\}$$

Then S is nonempty since there exists $x \in X$ and a supremum for $\{x\}$ is in S. Next we show that S is directed as follows. Any $y, z \in S$ are suprema of corresponding finite sets $Y \subseteq X$ and $Z \subseteq X$. Since $Y \cup Z$ is finite we get a supremum w of $Y \cup Z$ in S. It is easy to show that w is an upper bound of y and z.

Since $(\sqsubseteq)$ is directed complete, we obtain a supremum s for S. Then s is a supremum of X; here we only show that s is a bound of X. For any $x \in X$ we have a supremum x' of $\{x\}$ in S, and thus we have $x' \sqsubseteq s$. As $x \sqsubseteq x'$ by transitivity we conclude $x \sqsubseteq s$. ◀

The last argument in the above proof requires transitivity, but if we had reflexivity then x itself is a supremum of $\{x\}$ (see lemma extreme_bound_singleton) and so $x \sqsubseteq s$ would be immediate. Thus we can replace transitivity by reflexivity, but then pair-completeness does not imply finite completeness. We obtain the following result.

proposition (**in** reflexive) semicomplete_iff_directed_complete_finite_complete:
  shows "semicomplete (⊑) ⟷ directed_complete (⊑) ∧ finite_complete (⊑)"

We also tried to strengthen the above result by replacing finite completeness by pair completeness, but at the time of writing, the question is left open. We remark that, at least, Nitpick did not find a counterexample.

4 Knaster–Tarski-Style Fixed-Point Theorems

Given a monotone map $f : A \to A$ on a complete lattice $\langle A, \sqsubseteq\rangle$, the Knaster–Tarski theorem [21] states that

1. f has a fixed point in A, and
2. the set of fixed points forms a complete lattice.

Stauti and Maaden [20] generalized statement (1) where $\langle A, \sqsubseteq\rangle$ is a complete *trellis*—a complete pseudo-order—relaxing transitivity. They also proved a restricted version of (2), namely there exists a least (and by duality a greatest) fixed point in A.

In the following Section 4.1 we further generalize claim (1) so that any complete relation admits a *quasi-fixed point* $f(x) \sim x$, that is, $f(x) \sqsubseteq x$ and $x \sqsubseteq f(x)$. Quasi-fixed points are fixed points for antisymmetric relations; hence the Stauti–Maaden theorem is further generalized by relaxing reflexivity.

In Section 4.2 we also generalize claim (2) so that only a mild condition, which we call *attractivity*, is assumed. In this attractive setting quasi-fixed points are complete. Since attractivity is implied by either of transitivity or antisymmetry, in particular fixed points are complete in complete trellis, thus completing Stauti and Maaden's result.

In Section 4.3 we further generalize the result, proving that antisymmetry is sufficient for *strict* fixed points $f(x) = x$ to be complete.

4.1 Existence of Quasi-Fixed Points

First, we generalize the existence of fixed points so that nothing besides completeness is assumed on the relation. Fortunately, Quickcheck [3] quickly refutes the existence of *strict* fixed point $f(x) = x$ for an arbitrary complete relation.

▶ **Example 1** (by Quickcheck)**.** Let $A = \{\mathsf{a}_1, \mathsf{a}_2\}$, $(\sqsubseteq) = A \times A$, $f(\mathsf{a}_1) = \mathsf{a}_2$, and $f(\mathsf{a}_2) = \mathsf{a}_1$. Trivially f is monotone but $f(x) \neq x$ for either $x \in A$.

Hence, we instead show the existence of a quasi-fixed point $f(x) \sim x$. For reusability of proofs for the completeness results later on, we start with a stronger statement, namely: there exists a quasi-fixed point in any set of elements that is closed under f and complete for $(\sqsubseteq)$. Completeness restricted to a subset of elements is formalized as follows:

definition "complete_in S ≡ ∀X ⊆ S. Ex (extreme_bound_in S X)"

where predicate extreme_bound_in indicates the least elements among the bounds restricted to a given subset.

abbreviation "extreme_bound_in S X ≡ extreme (⊒) {b ∈ S. bound (⊑) X b}"

For convenience we construct a proof within the following context.

context
  fixes f **and** S
  assumes "monotone (⊑) (⊑) f" **and** "f ‘ S ⊆ S" **and** "complete_in (⊑) S"

Inspired by Stauti and Maaden [20], we start the proof by considering the set of subsets of S that are closed under f and themselves "complete":

definition AA **where** "AA ≡
  {A. A ⊆ S ∧ f ‘ A ⊆ A ∧ (∀B ⊆ A. ∀b. extreme_bound_in (⊑) S B b ⟶ b ∈ A)}"

Note here that by a "complete" subset $A \subseteq S$ we mean that *any* suprema with respect to S are in A, since suprema are not necessarily unique. We denote the intersection of all those subsets by C, and show that C contains a quasi-fixed point.

definition C **where** "C ≡ ⋂ AA"

lemma quasi_fixed_point_in_C: "∃c ∈ C. f c ∼ c"

Proof. We prove that any supremum c of C in S, which exists due to the completeness of S, is a quasi-fixed point of f. First, observe that $\mathsf{C} \in \mathsf{AA}$. Indeed:

- $\mathsf{C} \subseteq S$: since S is closed under f and complete, $S \in \mathsf{AA}$.
- $f(\mathsf{C}) \subseteq \mathsf{C}$: for every $A \in \mathsf{AA}$, we have $f(\mathsf{C}) \subseteq f(\mathsf{A}) \subseteq \mathsf{A}$. So $f(\mathsf{C}) \subseteq (\bigcap \mathsf{AA}) = \mathsf{C}$.
- completeness: given $B \subseteq \mathsf{C}$ and its supremum b in S, we prove $b \in \mathsf{C}$, that is, $b \in A'$ for every $A' \in \mathsf{AA}$. Indeed, we have $B \subseteq \mathsf{C} \subseteq A'$ and the definition of AA ensures $b \in A'$.

This implies that $c \in \mathsf{C}$. Moreover, since $f(\mathsf{C}) \subseteq \mathsf{C}$, we have $f(c) \in \mathsf{C}$, and since c is a supremum of C, we get $f(c) \sqsubseteq c$. It remains to prove the converse orientation $c \sqsubseteq f(c)$. To this end we consider the following set D:

define D **where** "D ≡ {x ∈ C. x ⊑ f c}"

We conclude by proving that $\mathsf{D} \in \mathsf{AA}$, since this implies $\mathsf{C} \subseteq \mathsf{D}$ and in particular $c \in \mathsf{D}$, which means $c \sqsubseteq f(c)$.

- $\mathsf{D} \subseteq S$: because $\mathsf{D} \subseteq \mathsf{C} \subseteq S$.
- $f(\mathsf{D}) \subseteq \mathsf{D}$: Let $d \in \mathsf{D}$. So $d \in \mathsf{C}$, and since c is a supremum of C, we have $d \sqsubseteq c$. With the monotonicity of f we get $f(d) \sqsubseteq f(c)$ and thus $f(d) \in \mathsf{D}$.
- completeness: Given $E \subseteq \mathsf{D}$ and its supremum b in S, we prove that $b \in \mathsf{D}$. Since $E \subseteq \mathsf{D}$, $f(c)$ is a bound of E, and as b is a least of such, $b \sqsubseteq f(c)$, that is $b \in \mathsf{D}$. ◀

By taking $S = \mathsf{UNIV}$ in the above lemma, we obtain:

theorem (**in** complete) monotone_imp_ex_quasi_fixed_point:
  assumes "monotone (⊑) (⊑) f" **shows** "∃s. f s ∼ s"

It is easy to see that this result indicates the existence of a strict fixed point if the relation ⊑ is antisymmetric, recovering statement (1) in the context of Stauti and Maaden [20], but without requiring reflexivity.

locale complete_antisymmetric = complete + antisymmetric

corollary (**in** complete_antisymmetric) monotone_imp_ex_fixed_point:
  assumes "monotone (⊑) (⊑) f" **shows** "∃s. f s = s"

4.2 Completeness of Quasi-Fixed Points

Next, we tacle the completeness of quasi-fixed points, generalizing statement (2). It was a surprise to us that, this time Nitpick [6] found a counterexample for this claim.

▶ **Example 2** (by Nitpick). We claimed (**in** complete) **assumes** "monotone (⊑) (⊑) f" **shows** "complete_in (⊑) {s. f s ∼ s}" and typed **nitpick**. In seconds it found a counterexample:

f = (λx. _) (a_1 := a_3, a_2 := a_3, a_3 := a_3, a_4 := a_1)
    (⊑) =
      (λx. _)
      (a_1 := (λx. _) (a_1 := False, a_2 := True, a_3 := True, a_4 := True),
       a_2 := (λx. _) (a_1 := True, a_2 := True, a_3 := True, a_4 := True),
       a_3 := (λx. _) (a_1 := True, a_2 := False, a_3 := True, a_4 := False),
       a_4 := (λx. _) (a_1 := True, a_2 := True, a_3 := True, a_4 := False))

Below we depict the relation ⊑ (left) and the mapping f (right).

On the left, arrow $a_i \rightarrow a_j$ means $a_i \sqsubseteq a_j$, and arrow $a_i \leftrightarrow a_j$ means $a_i \sim a_j$. On the right, an arrow $a_i \dashrightarrow a_j$ means $f(a_i) = a_j$. In this example, indeed ⊑ is complete and f is monotone. The quasi-fixed points are a_1, a_3, a_4; however, none of them are least, because $a_1 \not\sqsubseteq a_1$, $a_3 \not\sqsubseteq a_4$ and $a_4 \not\sqsubseteq a_4$.

After analysing the counterexample and existing proofs for lattices and trellises, we found a mild requirement on the relation ⊑, that we call *(semi)attractivity*:

locale semiattractive = less_eq_syntax +
  assumes attract: "x ⊑ y ⟹ y ⊑ x ⟹ x ⊑ z ⟹ y ⊑ z"

locale attractive = semiattractive + dual: semiattractive "(⊒)"

The intuition of this assumption is dipicted in Fig. 2. Attractivity is so mild that it is implied by either of antisymmetry and transitivity:

sublocale transitive ⊆ attractive **by** (unfold_locales, auto dest: trans)
sublocale antisymmetric ⊆ attractive **by** (unfold_locales, auto)

Figure 2 Attractivity: If two elements are similar, then arrows coming to one of them is also "attracted" to the other.

Assuming attractivity and completeness, we prove that the set of quasi-fixed points of a relation-preserving map f are complete. We start with a lemma saying that any complete subset S closed under f has a least quasi-fixed point:

```
lemma ex_extreme_quasi_fixed_point:
  assumes "monotone (⊑) (⊑) f" and "f ‘ S ⊆ S" and "complete_in (⊑) S"
    and attract: "∀q x. f q ∼ q ⟶ x ⊑ f q ⟶ x ⊑ q"
  shows "Ex (extreme (⊒) {q ∈ S. f q ∼ q})"
end
```

Proof. We start by defining the set of lower bounds of the quasi-fixed points in S.

```
define A where "A ≡ {a ∈ S. ∀s ∈ S. f s ∼ s ⟶ a ⊑ s}"
```

Let us first show that A ∈ AA, using the notation from the previous section.

- A $\subseteq S$: By definition.
- f(A) $\subseteq$ A: Let $a \in$ A. For any quasi-fixed point $s \in S$, we have that $a \sqsubseteq s$ and by monotonicity, $f(a) \sqsubseteq f(s)$. Since $f(s) \sim s$, by attract we get $f(a) \sqsubseteq s$, and thus $f(a) \in$ A.
- Completeness: Given $B \subseteq$ A, we show that any supremum b of B in S is in A. Since every quasi-fixed point s in S is a bound of A, s is a bound of B. As b is a least of such, we get $b \sqsubseteq s$ and thus $b \in$ A.

This implies C $\subseteq$ A, and with lemma quasi_fixed_point_in_C we obtain a quasi-fixed point in C $\subseteq$ A $\subseteq S$. This is a least one by the definition of A. ◀

Finally, we prove that the set of quasi-fixed points of f is complete.

```
locale complete_attractive = complete + attractive
```

```
theorem (in complete_attractive) monotone_imp_quasi_fixed_points_complete:
  assumes "monotone (⊑) (⊑) f" shows "complete_in (⊑) {s. f s ∼ s}"
```

Proof. Given a subset A of quasi-fixed points, we prove that A has a supremum *inside* the set of quasi-fixed points. Define S the set of bounds of A.

```
define S where "S ≡ {s. ∀a ∈ A. a ⊑ s}"
```

We prove that S satisfies the assumptions of ex_extreme_quasi_fixed_point:

- f(S) $\subseteq$ S: Let $s \in$ S. By the definition of S, for any $a \in A$ we have $a \sqsubseteq s$, and with monotonicity $f(a) \sqsubseteq f(s)$. Then by dual.attract with $f(a) \sim a$, we get $a \sqsubseteq f(s)$, and thus $f(s) \in$ S.
- Completeness: Due to the duality of completeness, it suffices to prove that every subset B of S has an infimum in S. As the universe is complete, B has an infimum b in UNIV. By the definition of S, every $a \in A$ is a lower bound of S and so of B. As b is a greatest of such, we get $a \sqsubseteq b$, concluding $b \in$ S.

Consequently, by ex_extreme_quasi_fixed_point, we find a least quasi-fixed point q in S. We conclude the proof by showing that q is a least bound of A, restricted to the set of quasi-fixed points:

- q is a quasi-fixed point: by construction.
- q is a bound of A: by construction, q is in S.
- q is least: Let p be another quasi-fixed point which is also a bound of A. Then p is a quasi-fixed point in S, and by construction of q, $q \sqsubseteq p$. ◀

The second result of Stauti and Maaden [20] states that, for a monotone map in a complete trellis, there exists a least fixed point. We have already obtained a stronger result: the set of fixed points are complete in complete trellises, since quasi-fixed points are precisely fixed points in pseudo-orders. Nevertheless, holding the as-general-as-possible manifesto in mind, we further generalize the result to show that antisymmetry alone is sufficient for the set of fixed points to be complete.

4.3 Completeness of Fixed Points in Antisymmetry

Now we prove that the set of strict fixed points is complete, only assuming antisymmetry. Observe first that this is not an immediate consequence of the completeness of quasi-fixed points, since when reflexivity is not available, there can be more fixed points than quasi-fixed points. So we have to show that there is no fixed points below the least quasi-fixed point we have found.

The proof relies on the following technical lemma, stating that given two sets A and B of strict fixed points, such that every element of A is below every element of B, there is a quasi-fixed point in-between.

```
lemma qfp_interpolant:
  assumes "complete (⊑)" and "monotone (⊑) (⊑) f"
    and "∀a ∈ A. ∀b ∈ B. a ⊑ b"
    and "∀a ∈ A. f a = a"
    and "∀b ∈ B. f b = b"
  shows "∃t. (f t ∼ t) ∧ (∀a ∈ A. a ⊑ t) ∧ (∀b ∈ B. t ⊑ b)"
```

Proof. We first define the set T of elements in between A and B:

```
define T where "T ≡ {t. (∀a ∈ A. a ⊑ t) ∧ (∀b ∈ B. t ⊑ b)}"
```

It is enough to prove that T satisfies the assumptions of lemma quasi_fixed_point_in_C:

- $f(\mathsf{T}) \subseteq \mathsf{T}$: Let $t \in \mathsf{T}$. Then for every $a \in A$, $a \sqsubseteq t$ and by monotonicity $f(a) \sqsubseteq f(t)$. Since a is a fixed point, we have $a = f(a) \sqsubseteq f(t)$. Similarly, we have $f(t) \sqsubseteq b$ for every $b \in B$, and thus $f(t) \in \mathsf{T}$.
- completeness: Let $C \subseteq \mathsf{T}$ and let us prove that C has a supremum in T. By the completeness of $(\sqsubseteq)$, we find a supremum c of $C \cup A$ in UNIV. Let us prove that this is a supremum of C in T:
  - $c \in \mathsf{T}$: By construction, c is a bound of A. Since $C \subseteq \mathsf{T}$, every $b \in B$ is a bound of C, and as c is least of such, $c \sqsubseteq b$. Consequently, $c \in \mathsf{T}$.
  - c is a bound of C: by construction.
  - c is least: Let $d \in \mathsf{T}$ be another bound of C. By the definition of T, d is also a bound of A, and so of $C \cup A$. As c is least of such, we conclude $c \sqsubseteq d$. ◀

From this lemma, we deduce that the set of strict fixed points is complete.

```
theorem (in complete_antisymmetric) monotone_imp_fixed_points_complete:
  assumes mono: "monotone (⊑) (⊑) f" shows "complete_in (⊑) {s. f s = s}"
```

Proof. Let A be a subset of strict fixed points. Similarly to the proof of attract_imp_qfp_complete, define the set S of bounds of A. This set S still satisfies the assumptions of ex_extreme_quasi_fixed_point, so it has a least *quasi*-fixed point q. We prove that this is a supremum of A with respect to the set of (strict) fixed points.

- q is a fixed point: by antisymmetry and the fact that q is a quasi-fixed point.
- q is a bound of A: because $q \in \mathsf{S}$.
- q is least: Let p be a fixed point and at the same time a bound of A. Let $B = \{q, p\}$. Then A and B satisfy the assumption of monotone_imp_interpolant_quasi_fixed_point. So there is a quasi-fixed point t between A and B. In particular, $t \sqsubseteq q$ and $t \sqsubseteq p$. Since t is a bound of A, we know $t \in \mathsf{S}$. Since q is a least quasi-fixed point in S, we get $q \sqsubseteq t$. With $t \sqsubseteq q$ and antisymmetry we get $q = t$, and since $t \sqsubseteq p$, we conclude $q \sqsubseteq p$. ◀

5 Kleene-Style Fixed-Point Theorems

Kleene's fixed-point theorem states that, for a pointed directed complete partial order $\langle A, \sqsubseteq\rangle$ and a Scott-continous map $f : A \to A$, the supremum of $\{f^n(\bot) \mid n \in \mathbb{N}\}$ exists in A and is a least fixed point. Mashburn [16] generalized the result so that $\langle A, \sqsubseteq\rangle$ is a ω-complete partial order and f is ω-continuous.

In this section we further generalize the result and show that for ω-complete relation $\langle A, \sqsubseteq\rangle$ and for every bottom element $\bot \in A$, the set $\{f^n(\bot) \mid n \in \mathbb{N}\}$ has suprema (not necessarily unique, of course) and, they are quasi-fixed points. Moreover, if $(\sqsubseteq)$ is attractive, then the suprema are precisely the least quasi-fixed points.

5.1 Scott Continuity, ω-Completeness, ω-Continuity

A related set $\langle A, \sqsubseteq\rangle$ is *ω-complete* if every ω-chain—a countable set in which any two elements are related—has a supremum. In order to characterize ω-chains in Isabelle (without going into ordinals), we model an ω-chain as the range of a relation-preserving map $c : \mathbb{N} \to A$.

```
locale omega_complete = less_eq_syntax +
  assumes "⋀ c :: nat ⇒ 'a. monotone (≤) (⊑) c ⟹ Ex (extreme_bound (⊑) (range c))"
```

A map $f : A \to A$ is *Scott-continuous* with respect to $(\sqsubseteq) \subseteq A \times A$ if for every directed subset $D \subseteq A$ with a supremum s, $f(s)$ is a supremum of the image $f(D)$.

```
definition "scott_continuous f ≡
  ∀D s. directed (⊑) D ⟶ extreme_bound (⊑) D s ⟶ extreme_bound (⊑) (f ` D) (f s)"
```

The notion of *ω-continuity* relaxes Scott-continuity by considering only ω-chain as D.

```
definition "omega_continuous f ≡ ∀c :: nat ⇒ 'a. ∀s.
  monotone (≤) (⊑) c ⟶
  extreme_bound (⊑) (range c) s ⟶ extreme_bound (⊑) (f ` range c) (f s)"
```

As $\langle \mathbb{N}, \leq\rangle$ is total, and thus directed, we can easily verify that Scott-continuity implies ω-continuity using the fact that the image of a monotone map over a directed set is directed.

```
lemma scott_continous_imp_omega_continous:
  assumes "scott_continuous f" shows "omega_continuous f"
```

For the later development we also prove that every ω-continuous function is *nearly* monotone, in the sense that it preserves relation $x \sqsubseteq y$ when x and y are reflexive elements. Note that near monotonicity coincides with monotonicity if the underlying relation is reflexive.

```
lemma omega_continous_imp_mono_refl:
  assumes "omega_continuous f" and "x ⊑ y" and "x ⊑ x" and "y ⊑ y"
  shows "f x ⊑ f y"
```

Proof. The proof consists in observing that under the assumptions, function c :: nat ⇒ 'a defined by "c i ≡ if i = 0 then x else y" is monotone. Furthermore, y is a supremum of the image of c, i.e., $\{x, y\}$, so ω-continuity ensures that $f(y)$ is a supremum of $\{f(x), f(y)\}$, which in particular means that $f(x) \sqsubseteq f(y)$. ◀

5.2 Kleene's Fixed-Point Theorem

The first part of Kleene's theorem demands to prove that the set $\{f^n(\bot) \mid n \in \mathbb{N}\}$ has a supremum and that all such are quasi-fixed points. We prove this claim without assuming anything on the relation $\sqsubseteq$ besides ω-completeness and one bottom element.

```
context
  fixes f and bot ("⊥")
  assumes "omega_complete (⊑)" and "omega_continuous (⊑) f" and "∀x. ⊥ ⊑ x"
begin
```

Just for convenience we abbreviate the set $\{f^n(\bot) \mid n \in \mathbb{N}\}$ as Fn in Isabelle:

```
abbreviation(input) fn where "fn n ≡ (f ^^ n) ⊥"
abbreviation(input) "Fn ≡ range fn"
```

```
theorem kleene_quasi_fixed_point:
  shows "∃p. extreme_bound (⊑) Fn p" and "extreme_bound (⊑) Fn p ⟹ f p ∼ p"
```

Proof. First note that fn is a relation-preserving map from $\langle \mathbb{N}, \leq \rangle$ to $\langle A, \sqsubseteq \rangle$: this is reduced to $f^n(\bot) \sqsubseteq f^{n+k}(\bot)$ for any n and k, which is easily proved by induction on n. Thus Fn = range fn is an ω-chain, and ω-completness gives a supremum, say p, for Fn. Now let us prove that p is a quasi-fixed point.

Since p is a supremum of Fn, the ω-continuity of f ensures that $f(p)$ is a supremum of $f(\mathsf{Fn})$. As p is a bound of Fn, it is also a bound of $f(\mathsf{Fn})$ due to the definition of Fn. Consequently, $f(p) \sqsubseteq p$.

It remains to show the other orientation $p \sqsubseteq f(p)$. Since p is least in the bounds of Fn, it suffices to show that $f(p)$ is a bound of Fn, that is, $f^n(\bot) \sqsubseteq f(p)$ for every n. We prove this by induction on n. The base case is by the assumption of $\bot$. For inductive case, assume $f^n(\bot) \sqsubseteq p$. By the "near" monotonicity we conclude $f^{n+1}(\bot) \sqsubseteq f(p)$, but to this end we need $f^n(\bot) \sqsubseteq f^n(\bot)$ for every n, which would be trivial if we had reflexivity. Instead we prove this fact by induction on n, also using omega_continous_imp_mono_refl. ◀

Now the first part of Kleene's theorem is reproved without any order assumption: for an ω-complete set $\langle A, \sqsubseteq \rangle$ with a bottom element $\bot$ and ω-continuous map $f : A \to A$, there exists a supremum for $\{f^n(\bot) \mid n \in \mathbb{N}\}$ and it is a quasi-fixed point.

Kleene's theorem also states that the quasi-fixed point found this way is a least one. Hence naturally we consider proving this claim for arbitrary relations, but again Nitpick saved us this hopeless effort.

▶ **Example 3** (by Nitpick). Our conjecture is now "extreme_bound (⊑) Fn q ⟹ extreme (⊒) {s. f s ∼ s} q". Following is a counterexample found by Nitpick:

```
⊥ = a₁
    f = (λx. _) (a₁ := a₃, a₂ := a₁, a₃ := a₃)
    (⊑) =
      (λx. _)
      (a₁ := (λx. _) (a₁ := True, a₂ := True, a₃ := True),
       a₂ := (λx. _) (a₁ := True, a₂ := False, a₃ := True),
       a₃ := (λx. _) (a₁ := True, a₂ := False, a₃ := True))
    q = a₃
```

In this example, indeed a_1 is a bottom element, $\sqsubseteq$ is (ω-)complete, and f is ω-continuous. The set of quasi-fixed points is $\{\mathsf{a}_1, \mathsf{a}_2, \mathsf{a}_3\}$, and a_3 is an extreme bound of $\{\mathsf{f}^n(\bot) \mid n \in \mathbb{N}\} = \{\mathsf{a}_1, \mathsf{a}_3\}$. However, a_3 is not a least quasi-fixed point because $\mathsf{a}_3 \not\sqsubseteq \mathsf{a}_2$.

Now again, attractivity turns out to be the key. We prove that the set of suprema of Fn coincides with the set of least quasi-fixed points, if the underlying relation is attractive.

corollary (**in** attractive) kleene_fixed_point_dual_extreme:
  shows "extreme_bound (⊑) Fn = extreme (⊒) {s. f s ∼ s}"

Proof. Let q be a supremum of Fn. By kleene_quasi_fixed_point, we already know that this is a quasi-fixed point. So to prove that q is a least quasi-fixed point, it is enough to show that any other quasi-fixed point s is a bound of $\mathsf{Fn} = \{f^n(\bot) \mid n \in \mathbb{N}\}$. This is done by induction on n. The base case $\bot \sqsubseteq s$ is trivial by assumption. For the inductive case, assuming $f^n(\bot) \sqsubseteq s$ we get $f^{n+1}(\bot) \sqsubseteq f(s)$ by the same argument as in the previous proof. Since $f(s) \sim s$, attractivity concludes $f^{n+1}(\bot) \sqsubseteq s$.

Conversely, consider a least quasi-fixed point s. We show that s is a supremum of Fn. Since s is a quasi-fixed point, and as we have just proved above, s is a bound of Fn. It remains to prove that s is least in bounds of Fn.

By kleene_quasi_fixed_point, Fn has a supremum, say k, and is a quasi-fixed point. As s is a least quasi-fixed point, we have $s \sqsubseteq k$. On the other hand, as s is a bound of Fn and k is a least of such, we see $k \sqsubseteq s$. Consequently, $s \sim k$.

Now let x be a bound of Fn. We know $k \sqsubseteq x$, and with $s \sim k$, we conclude $s \sqsubseteq x$ due to attractivity. ◀

6 Conclusion

In this paper, we developed an Isabelle/HOL formalization for order-theoretic concepts such as various completeness conditions and fixed-point theorems. We adopt an as-general-as-possible approach, so that many results previously known only for partial orders or pseudo-orders are generalized. In particular the generalizations of the Knaster–Tarski theorem and Kleene's fixed-point theorems would deserve some attention. These achievement become reachable to us largely due to the great assistance by the smart Isabelle 2018 environment.

For future work, it is tempting to further formalize and hopefully generalize other results about completeness and fixed points, which are listed as related work in the introduction. We also plan to extend the library with convergence arguments, which were actually our original motivation for formalizing these order-theoretic concepts.

References

1. Samson Abramsky and Achim Jung. *Domain Theory*. Number III in Handbook of Logic in Computer Science. Oxford University Press, 1994.
2. Clemens Ballarin. Interpretation of locales in Isabelle: Theories and proof contexts. In Jonathan M. Borwein and William M. Farmer, editors, *Proceedings of the 5th International Conference on Mathematical Knowledge Management (MKM 2006)*, volume 4108 of *LNCS*, pages 31–43. Springer Berlin Heidelberg, 2006. doi:10.1007/11812289_4.
3. Stefan Berghofer and Tobias Nipkow. Random testing in Isabelle/HOL. In *Proceedings of the 2nd International Conference on Software Engineering and Formal Methods (SEFM 2004)*, pages 230–239. IEEE Computer Society, 2004. doi:10.1109/SEFM.2004.36.
4. George M. Bergman. *An Invitation to General Algebra and Universal Constructions*. Springer International Publishing, Cham, 2015. doi:10.1007/978-3-319-11478-1.
5. S. Parameshwara Bhatta. Weak chain-completeness and fixed point property for pseudo-ordered sets. *Czechoslovak Mathematical Journal*, 55(2):365–369, 2005.
6. Jasmin Christian Blanchette and Tobias Nipkow. Nitpick: A counterexample generator for higher-order logic based on a relational model finder. In Matt Kaufmann and Lawrence C. Paulson, editors, *Proceedings of the 1st International Conference on Interactive Theorem Proving (ITP 2010)*, volume 6172 of *LNCS*, pages 131–146. Springer Berlin Heidelberg, 2010. doi:10.1007/978-3-642-14052-5_11.
7. Sascha Böhme and Tobias Nipkow. Sledgehammer: Judgement day. In *Proceedings of the 5th International Joint Conference on Automated Reasoning (IJCAR 2010)*, volume 6173 of *LNCS*, pages 107–121. Springer Berlin Heidelberg, 2010. doi:10.1007/978-3-642-14203-1_9.
8. B. A. Davey and H. A. Priestley. *Introduction to Lattices and Order*. Cambridge University Press, 2002. doi:10.1017/CBO9780511809088.
9. G. Gierz, K. H. Hofmann, K. Keimel, J. D. Lawson, M. W. Mislove, and D. S. Scott. *Continuous Lattices and Domains*. Cambridge University Press, 2003. doi:10.1017/CBO9780511542725.
10. Georges Gonthier. Formal proof – the four-color theorem. *Notices of the AMS*, 55(11):1382–1393, 2008.
11. Florian Haftmann and Tobias Nipkow. A code generator framework for Isabelle/HOL. In Klaus Schneider and Jens Brandt, editors, *Theorem Proving in Higher Order Logics: Emerging Trends*, pages 128–143. Department of Computer Science, University of Kaiserslautern, 2007.
12. Thomas Hales, Mark Adams, Gertrud Bauer, Tat Dat Dang, John Harrison, Hoang Le Truong, Cezary Kaliszyk, Victor Magron, Sean McLaughlin, Tat Thang Nguyen, et al. A formal proof of the Kepler conjecture. *Forum of Mathematics, Pi*, 5:e2, 2017. doi:10.1017/fmp.2017.1.
13. Florian Kammüller. Modular reasoning in Isabelle. In David McAllester, editor, *Proceedings of the 17th International Conference on Automated Deduction (CADE-17)*, volume 1831 of *LNCS*, pages 99–114. Springer Berlin Heidelberg, 2000. doi:10.1007/10721959_7.
14. Gerwin Klein, Kevin Elphinstone, Gernot Heiser, June Andronick, David Cock, Philip Derrin, Dhammika Elkaduwe, Kai Engelhardt, Rafal Kolanski, Michael Norrish, Thomas Sewell, Harvey Tuch, and Simon Wiwood. seL4: Formal verification of an OS kernel. In *Proceedings of the ACM SIGOPS 22nd Symposium on Operating Systems Principles (SOSP 2009)*, pages 207–220. ACM, 2009. doi:10.1145/1629575.1629596.
15. K. Leutola and J. Nieminen. Posets and generalized lattices. *Algebra Universalis*, 16(1):344–354, 1983.
16. J. D. Mashburn. The least fixed point property for omega-chain continuous functions. *Houston Journal of Mathematics*, 9(2):231–244, 1983.

17 T. Nipkow, L.C. Paulson, and M. Wenzel. *Isabelle/HOL – A Proof Assistant for Higher-Order Logic*, volume 2283 of *LNCS*. Springer, 2002. doi:10.1007/3-540-45949-9.

18 S. Parameshwara Bhatta and Shiju George. Some fixed point theorems for pseudo ordered sets. *Algebra and Discrete Mathematics*, 11(1):17–22, 2011.

19 H.L. Skala. Trellis theory. *Algebra Univ.*, 1:218–233, 1971. doi:10.1007/BF02944982.

20 Abdelkader Stouti and Abdelhakim Maaden. Fixed points and common fixed points theorems in pseudo-ordered sets. *Proyecciones*, 32(4):409–418, 2013. doi:10.4067/S0716-09172013000400008.

21 Alfred Tarski. A lattice-theoretical fixpoint theorem and its applications. *Pacific Journal of Mathematics*, 5(2):285–309, 1955.

22 Makarius Wenzel. Isabelle/jEdit – a prover IDE within the PIDE framework. In *Proceedings of the 5th Conferences on Intelligent Computer Mathematics (CICM 2012)*, volume 7362 of *LNCS*, pages 468–471. Springer Berlin Heidelberg, 2012. doi:10.1007/978-3-642-31374-5_38.